

# Lenguaje de Diseño en PSeInt

## Subalgoritmos

RESOLUCIÓN DE PROBLEMAS  
FUNDAMENTOS DE LA PROGRAMACIÓN

*Ingeniería en Informática*  
*Ingeniería en Computación*  
*Ingeniería en Minas*  
*Profesorado en Computación*



UNIVERSIDAD NACIONAL DE SAN LUIS  
DEPARTAMENTO DE INFORMÁTICA  
AÑO 2025

# Índice

<b>1. Introducción</b>	<b>2</b>
<b>2. Definición de Subalgoritmos en PSeInt</b>	<b>3</b>
<b>3. Definición, Invocación y Retorno de Subalgoritmos</b>	<b>7</b>
<b>4. ANEXO: Ejemplos en PSeInt</b>	<b>8</b>
4.1. Ejemplo 1: . . . . .	8
4.2. Ejemplo 2: . . . . .	9



## Subalgoritmos

### 1. Introducción

Dado un problema relativamente complejo, la manera más conveniente de resolverlo es dividirlo en tareas o módulos de menor complejidad. A estos módulos, en lenguaje de diseño se los denomina **subalgoritmos**, **subprogramas** o **subprocesos**.

Un subalgoritmo tiene que ser **lógicamente coherente** y **autocontenido** y formar parte de un algoritmo de mayor envergadura. Por **lógicamente coherente** se quiere significar que contiene un conjunto determinado de acciones que conducen a la solución de una tarea específica, relativamente simple; mientras que **autocontenido** significa que es independiente de cualquier otro problema y de sus soluciones. Los subalgoritmos son una facilidad del lenguaje de diseño (y de los lenguajes de programación) que permiten que determinados conjuntos de acciones puedan ser provistos en una forma modular. La **característica** de este conjunto de acciones es que realizan tareas comunes que pueden utilizarse para resolver distintos tipos de problemas o bien, varias veces dentro de un mismo problema, pero con distintos datos.

Estos subalgoritmos se escriben una vez y, luego, son usados por todos aquellos que requieran de ellos. Por ejemplo: ordenar datos en orden ascendente/descendente, como así también las funciones primitivas propias del lenguaje como por ejemplo *valor absoluto* (abs), *raíz cuadrada* (rc), *número aleatorio* (azar), entre otros.

Lo expresado puede verse gráficamente en la Figura 1:




Figura 1

En esta figura los subalgoritmos M, N, P, O y Q han sido escritos una vez y son usados indistintamente por

los algoritmos A y B, o entre ellos.

## 2. Definición de Subalgoritmos en PSeInt

En PSeInt hay dos tipos de subprocesos:

**Funciones:** realizan un proceso y devuelven un resultado. Por ejemplo, una función puede sumar dos números y retornar el resultado.

**Procedimientos:** también realizar un proceso, pero no necesariamente van a retornar un resultado. En general, solo realizan una tarea específica.

Los subalgoritmos se definen colocando la palabra reservada **SUBALGORITMO**.

### SINTÁXIS DE DECLARACIÓN DE FUNCIONES

Los subalgoritmos que van a ser utilizados como **funciones** (es decir, devuelven un resultado) se definen colocando la palabra reservada **SubAlgoritmo** seguida de la variable de retorno, la especificación de su nombre y, opcionalmente, los argumentos o parámetros (los términos *argumento* y *parámetro* son sinónimos). En el cuerpo de la función se realiza la declaración de sus propias variables (*variables locales*) y la secuencia de acciones que hacen a la logística propia del mismo.

Es obligatorio **declarar** la variable de retorno y **asignarle** algún valor y solamente se puede definir una única variable de retorno, es decir, la función devuelve un único resultado.

### Sintaxis de definición de una Función

**SubAlgoritmo** <Nombre de la variable de retorno> ← *Nombre* (<Nombre del parámetro 1>, <Nombre del parámetro 2>, ...)

*Declaracion de variables*

*Secuencia de acciones*

**FinSubAlgoritmo**

El nombre de un subalgoritmo se construye siguiendo las mismas restricciones que para las variables y los arreglos.



Los parámetros en un subalgoritmo son canales de comunicación entre quien invoca y el módulo invocado. Los parámetros que aparecen en la definición de un subalgoritmo se conocen con el nombre de **parámetros formales** y **se utilizan sólo** en las **acciones** que conforman el **cuerpo** del subalgoritmo, al igual que las variables locales que allí se definan. Estos parámetros permiten ingresar datos al subproceso u obtener resultados desde el mismo y que serán comunicados al módulo que lo invocó. Para utilizar el subalgoritmo es necesario invocarlo a través del nombre y la lista de argumentos necesarios para que el mismo funcione. Los parámetros utilizados en el momento de la invocación reciben el nombre de **parámetros actuales o reales**. El número de parámetros actuales debe ser igual al número de parámetros formales.

Pueden definirse dos clases diferentes de parámetros formales:

- **Por valor** son sólo **parámetros de entrada**. Implica que el argumento formal recibe una copia del contenido de la variable (o el resultado de una expresión) que se utilizó al momento de realizar la llamada o **invocación** del subprograma. Las modificaciones que se realicen a esos valores en el cuerpo del módulo no se reflejarán fuera del mismo. Si el parámetro formal es una variable simple y no se define el tipo de pasaje, se asume que es por valor.
- **Por referencia** son **parámetros de entrada y de salida**, es decir el subalgoritmo puede devolver resultados de la ejecución de sus acciones a quien lo invocó. Tanto el parámetro formal como la variable utilizada al momento de invocar al subalgoritmo (parámetro actual) comparten la misma posición de memoria. Es por eso que, cualquier cambio que se realice al parámetro formal en el cuerpo del módulo, se refleja directamente en el respectivo parámetro actual. Si el parámetro formal es un arreglo y no se define el tipo de pasaje, se asume que es por referencia.

## Ejemplo de Función

**SubAlgoritmo** *EsnoES*  $\leftarrow$  Es-raiz-cuadrada (*xx*, *yy*)

*Definir EsnoEs como logico*

*EsnoEs*  $\leftarrow$  FALSO

*Si xx = rc(yy) entonces*

*EsnoEs*  $\leftarrow$  VERDADERO

*FinSi*

**FinSubAlgoritmo**

Donde:



- *EsnoEs* es la **variable de retorno** que recibirá el resultado de ejecutar la función.
- *Es-raiz-cuadrada* es el **nombre** de la función.
- *xx e yy* son los **parámetros formales** y el pasaje es por **valor**

#### Algoritmo Raiz-Cuadrada

```
Definir A,B como enteros
Definir Respuesta como logico
Escribir “Ingresar valor”
Leer A
Escribir “Ingresar valor”
Leer B
// Invocacion de la Funcion Es-raiz-cuadrada
Respuesta ← Es-raiz-cuadrada(A, B)

Si Respuesta = VERDADERO entonces
    Escribir A, “es la raiz cuadrada de ”, B
FinSi
```

#### FinAlgoritmo

Donde:

- *Respuesta* es la **variable** que recibirá el resultado de invocar a la función.

Los subalgoritmos que van a ser utilizados como **procedimientos** NO necesitan de una variable de retorno. Se definen colocando la palabra reservada **SubAlgoritmo** seguida de la especificación de su nombre y, opcionalmente, los parámetros que sean necesarios. En el cuerpo del subproceso se realiza la declaración de sus propias variables (*variables locales*) y la secuencia de acciones que hacen a la logística propia del mismo. No se **declara** variable de retorno y en los casos que sea necesario que el procedimiento retorne resultados a quien lo invocó, **deberá** hacerlo a través del pasaje de parámetros por referencia. De esta manera podrá retornar más de un resultado.

## Sintaxis de definición de un Procedimiento

**SubAlgoritmo** *Nombre* (<Nombre del parámetro 1>, <Nombre del parámetro 2>, ...)

*Declaracion de variables*

*Secuencia de acciones*

**FinSubAlgoritmo**

## Ejemplo de Procedimiento

**SubAlgoritmo** Factorial (N, FAC por referencia)

Definir *I* como entero

FAC ← 1

**Para** *I* ← 1 **Hasta** *N* **Con Paso 1 Hacer**

FAC ← FAC\**I*

**FinPara**

**FinAlgoritmos**

**Algoritmo** Factorial-de-un-numero

// Este programa calcula el factorial del valor N, N<sup>2</sup> y N<sup>3</sup>

Definir N,RESUL como enteros

**Escribir** "Ingrese el valor al que le desea calcular el factorial"

**Leer** *N*

// Invocacion del Procedimiento Factorial para N

**Factorial** (N, RESUL)

**Escribir** "El factorial de ", N "es", RESUL

// Invocacion del Procedimiento Factorial para N<sup>2</sup>

**Factorial** (N\*N, RESUL)

**Escribir** "El resultado es", RESUL

// Invocacion del Procedimiento Factorial para N<sup>3</sup>

**Factorial** (N<sup>3</sup>, RESUL)

**Escribir** "El resultado es", RESUL

**FinAlgoritmo**



En el ejemplo del subalgoritmo FACTORIAL, el parámetro  $N$  es un parámetro por valor (no se ha especificado ningún tipo de pasaje de parámetro), que le provee al subalgoritmo el valor del número para el cual se debe calcular el factorial, mientras que el parámetro FAC es un parámetro de salida (por referencia) y es el argumento que devuelve al módulo que lo invocó el resultado del cálculo del factorial.

Notar que en la segunda y tercera invocación de FACTORIAL se ha pasado como primer argumento una expresión. Esta pasaje es válido debido a que el primer parámetro ha sido definido por valor.

Las variables que se declaran dentro de un subalgoritmo sólo pertenecen al ambiente del subalgoritmo donde están definidas y se conocen con el nombre de **variables locales**. Tienen el mismo ámbito de uso que los parámetros formales del subalgoritmo. En este ejemplo es el caso de la variable  $I$ . El **ámbito** se define entre las palabras reservadas **SubAlgoritmo** y **FinSubAlgoritmo** y la secuencia de acciones comprendidas entre dichas sentencias forman parte del cuerpo del subalgoritmo.

### 3. Definición, Invocación y Retorno de Subalgoritmos

Todos los subalgoritmos o subprogramas deben definirse antes del algoritmo principal y antes de ser invocados. Se **invoca** al subalgoritmo a través de su nombre, seguido de la lista de argumentos actuales, los cuales deben coincidir en cantidad y orden con los que el mismo fue definido. Para que las acciones descritas en el subalgoritmo sean ejecutadas, es necesario que el mismo **sea invocado desde el algoritmo principal ó desde otro subalgoritmo**.

Es importante aclarar que si el subalgoritmo fue definido con una determinada cantidad de parámetros formales, debe ser invocado con esa misma cantidad de parámetros actuales.

Cada vez que un subalgoritmo es invocado se establece, automáticamente, una **correspondencia** entre los parámetros formales y los actuales. Esta correspondencia está definida **por la posición que los parámetros ocupan dentro de la lista de parámetros**, es decir, el primer parámetro actual se corresponde con el primer parámetro formal; el segundo parámetro actual con el segundo parámetro formal y, así siguiendo, hasta completar todos los parámetros.

Cuando se invoca a un subalgoritmo, **la ejecución del módulo que invoca se suspende y el control de la ejecución se transfiere a la primera acción del cuerpo del subalgoritmo invocado**. Luego de la ejecución de las acciones que describen la lógica del cuerpo del subalgoritmo, el **control de la ejecución** retorna a la sentencia siguiente a la invocación del subalgoritmo en el módulo invocante.

Para los **parámetros formales** que fueron definidos:

- **por valor:** los **parámetros actuales** pueden ser constantes, variables (definidas en el ambiente del módu-



lo invocante), expresiones o valores de funciones de PSeInt.

- **por referencia:** los **parámetros actuales** deben ser variables simples o arreglos definidos en el ambiente del módulo invocante, pues allí el subalgoritmo devuelve sus resultados (tanto el parámetro formal como el actual comparten las posiciones de memoria asignadas a esas variables).

## 4. ANEXO: Ejemplos en PSeInt

### 4.1. Ejemplo 1:

¿Qué resuelve el siguiente programa?. Redacte el enunciado.

```
1 // función que recibe un argumento por valor, y devuelve su doble
2 SubAlgoritmo res ← CalcularDoble(num)
3   Definir res como real
4   res ← num*2 // retorna el doble
5 FinSubAlgoritmo
6
7 // procedimiento que recibe un argumento por referencia, y lo modifica
8 SubAlgoritmo Triplicar(num por referencia)
9   num ← num*3 // modifica la variable duplicando su valor
10 FinSubAlgoritmo
11 // procedimiento que recibe un argumento por referencia y otro por valor,
12 //y modifica al argumento por referencia
13 SubAlgoritmo es_multiplo(esnoes por referencia, num1 por valor, num2 por valor)
14   Definir Resto como entero
15   Resto ← num1%num2 // modifica la variable duplicando su valor
16   esnoes← FALSO
17   Si (Resto = 0) Entonces
18     esnoes← VERDADERO
19   FinSi
20 FinSubAlgoritmo
```



```
22 // proceso principal, que invoca a las funciones antes declaradas
23
24 Algoritmo PruebaFunciones
25     Definir x Como Real
26     Definir esmultiplo como logico
27     Definir N1, N2 como enteros
28     Escribir "Ingrese un valor numérico para x:"
29     Leer x
30     Escribir "Llamada a la función CalcularDoble (pasaje por valor)"
31     Escribir "El doble de ",x," es ", CalcularDoble(x)
32     Escribir "El valor original de x es ",x
33     Escribir "Llamada al procedimiento Triplicar (pasaje por referencia)"
34     Triplicar(x)
35     Escribir "El nuevo valor de x es ", x
36     Escribir "Llamada al procedimiento esnoes (pasaje por referencia y pasaje por valor)"
37     Escribir "Ingrese un valor"
38     Leer N1
39     Escribir "Ingrese un valor"
40     Leer N2
41     Escribir "Llamada al procedimiento es_multiplo (pasaje por referencia y por valor)"
42     esmultiplo← FALSO
43     es_multiplo(esmultiplo, N1, N2)
44     Escribir "N1 es multiplo de N2? ", esmultiplo
45
46 FinAlgoritmo
```

## 4.2. Ejemplo 2:

El dueño de un gimnasio quiere hacer un estudio sobre los planes de sus socios. Por cada cliente se tienen los siguientes datos: el tipo de actividad del plan contratado (aparatos, fitness o natación) y la cantidad de clases semanales de las que dispone (2, 4 o libre). Se pide: Diseñar un procedimiento que permita al usuario consultar por la cantidad de socios que han contratado un tipo de actividad en particular especificando, además la cantidad de clases semanales que disponen.



```

1 SubAlgoritmo ingreso(socios2,li2, ls2 por referencia)
2   definir i Como Entero
3   Escribir "Ingrese la cantidad de socios que quiere cargar"
4   Leer ls2
5   Mientras ls2 < 1 ∨ ls2 >100 Hacer
6     Escribir "Ingrese la cantidad de socios que quiere cargar, entre 1 y 100"
7     Leer ls2
8   FinMientras
9   Para i ← li2 Hasta ls2*2 con paso 2 Hacer
10    Escribir "Ingrese el tipo de actividad: 1.Aparatos, 2.Fitness, 3.Natación"
11    Leer socios2[i]
12    Mientras socios2[i] ≠ 1 ∧ socios2[i] ≠ 2 ∧ socios2[i] ≠ 3 Hacer
13      Escribir "Ingrese el tipo de actividad correctamente: 1.Aparatos, 2.Fitness, 3.Natación"
14      Leer socios2[i]
15    FinMientras
16    Escribir "Ingrese la cantidad de clases: 1.2 veces por semana, 2.4 veces por semana, 3.Libre"
17    Leer socios2[i+1]
18    Mientras socios2[i+1] ≠ 1 ∧ socios2[i+1] ≠ 2 ∧ socios2[i+1] ≠ 3 Hacer
19      Escribir "Ingrese la cantidad de clases correctamente: 1.2 veces por semana, 2.4 veces por semana, 3.Libre"
20      Leer socios2[i+1]
21    FinMientras
22  FinPara
23 FinSubAlgoritmo

25 SubAlgoritmo consultar(socios2,li2, ls2, tipo2, clases2, cantidad2 por referencia )
26   definir i Como Entero
27   cantidad2 ← 0
28   Para i ← li2 Hasta ls2*2 con paso 2 Hacer
29     Si socios2[i] == tipo2 ∧ socios2[i+1] == clases2 Entonces
30       cantidad2 ← cantidad2 +1
31     FinSi
32   FinPara
33 FinSubAlgoritmo

35 Algoritmo P7E5
36 Definir tipo, clases, res,li,ls Como Entero
37 Definir socios Como Entero
38 Dimension socios[200]
39 li← 1
40 ingreso(socios,li,ls)
41 Escribir "Ingrese el tipo de actividad a consultar : 1.Aparatos, 2.Fitness, 3.Natación"
42 Leer tipo
43 Mientras tipo ≠ 1 ∧ tipo ≠ 2 ∧ tipo ≠ 3 Hacer
44   Escribir "Ingrese el tipo de actividad a consultar correctamente: 1.Aparatos, 2.Fitness, 3.Natación"
45   Leer tipo
46 FinMientras
47 Escribir "Ingrese la cantidad de clases a consultar: 1.2 veces por semana, 2.4 veces por semana, 3.Libre"
48 Leer clases
49 Mientras clases ≠ 1 ∧ clases ≠ 2 ∧ clases ≠ 3 Hacer
50   Escribir "Reingrese la cantidad de clases a consultar: 1.2 veces por semana, 2.4 veces por semana, 3.Libre"
51   Leer clases
52 FinMientras
53 consultar(socios, li, ls, tipo, clases, res)
54 Escribir "La cantidad de clientes con el tipo de actividad", tipo, "y la cantidad de clases", clases, "es: ", res
55 FinAlgoritmo

```

